
capture log close
set more off
set matsize 10000

use rep_final, clear

	*noncitizens
	keep if cit==0
	*married last year, currently still married with spouse present
	keep if marrinyr==2 & married==1 
	*spouse info is avail.
	drop if citizen_sp==.


* Define control variables 

	local demog "age age2 male yreduc i.racegr"
	local imm "ysm i.bpl " 
	local fe "i.statefip i.yrmarr "
		
	local full "`demog' `imm' `fe' "

	local interactpost "c.age##post c.age2##post i.male##post c.yreduc##post i.racegr##post c.ysm##post i.bpl##post i.statefip##post i.yrmarr"


	
* Use yrmarr to measure time treatment because individuals are surveyed throughout the year
	
	g post=(yrmarr>=2013)
	g post1=(yrmarr>=2013 & yrmarr<=2016)
	g post2=(yrmarr>=2017)
	
	g treatpost=treatgr*post
	g treatpost1=treatgr*post1
	g treatpost2=treatgr*post2
	
	
	local a=1
	
	foreach controlgr of varlist controlgrdaca controlgrdacacutoff controlgrdaca_agematch ///
								controlgrchildimm controlgrdacahs  {
		
		foreach depvar of varlist cit_sp  {
	
			
			display ""
			display "--------------------------------------"
			
			display "Dependent Variable: `depvar' "
			display "Control group: `controlgr' "		
			
			display "--------------------------------------"
		
			reg `depvar' treatpost1 treatpost2 treatgr `interactpost'  if daca==1|`controlgr'==1   [pweight = perwt] , cluster(statefip)
			est sto e`a'			
			
			local a=`a'+1
			}
			
		*dep var mean
	
		sum cit_sp [fweight=perwt] if e(sample) 
		
		}
			*Make table 
	
			esttab e1 e2 e3 e4 e5 ///
					using tableA6.csv, se  ///
					mtitles("DACA-Ineligible Undocumented Immigrants" ///
							"DACA-Ineligible Missing Cutoffs" ///
							"DACA-Ineligible Age Matched" ///
							"DACA-Ineligible Childhood Arrivals" ///
							"DACA-Ineligible with HS/GED" ) ///
					title(Intermarriage Rate) ///
					b(%9.4f) se(%9.4f) star(* 0.1 ** 0.05 *** 0.01) ///
					nogaps replace
			
			